<commit_before># Module for the IPv6 Router Advertisement Daemon.

{ config, lib, pkgs, ... }:

with lib;

let

  cfg = config.services.radvd;

  confFile = pkgs.writeText "radvd.conf" cfg.config;

in

{

  ###### interface

  options = {

    services.radvd.enable = mkOption {
      default = false;
      description =
        ''
          Whether to enable the Router Advertisement Daemon
          (<command>radvd</command>), which provides link-local
          advertisements of IPv6 router addresses and prefixes using
          the Neighbor Discovery Protocol (NDP).  This enables
          stateless address autoconfiguration in IPv6 clients on the
          network.
        '';
    };

    services.radvd.config = mkOption {
      example =
        ''
          interface eth0 {
            AdvSendAdvert on;
            prefix 2001:db8:1234:5678::/64 { };
          };
        '';
      description =
        ''
          The contents of the radvd configuration file.
        '';
    };

  };


  ###### implementation

  config = mkIf cfg.enable {

    users.extraUsers.radvd =
      { uid = config.ids.uids.radvd;
        description = "Router Advertisement Daemon User";
      };

    systemd.services.radvd =
      { description = "IPv6 Router Advertisement Daemon";

        wantedBy = [ "multi-user.target" ];

        after = [ "network.target" ];

        path = [ pkgs.radvd ];

        preStart = ''
          mkdir -m 755 -p /run/radvd
          chown radvd /run/radvd
        '';

        serviceConfig =
          { ExecStart = "@${pkgs.radvd}/sbin/radvd radvd"
              + " -p /run/radvd/radvd.pid -m syslog -u radvd -C ${confFile}";
            Restart = "always";
            Type = "forking";
            PIDFile = "/run/radvd/radvd.pid";
          };
      };

  };

}<commit_msg>radvd service: fix due to systemd upgrade<commit_after># Module for the IPv6 Router Advertisement Daemon.

{ config, lib, pkgs, ... }:

with lib;

let

  cfg = config.services.radvd;

  confFile = pkgs.writeText "radvd.conf" cfg.config;

in

{

  ###### interface

  options = {

    services.radvd.enable = mkOption {
      default = false;
      description =
        ''
          Whether to enable the Router Advertisement Daemon
          (<command>radvd</command>), which provides link-local
          advertisements of IPv6 router addresses and prefixes using
          the Neighbor Discovery Protocol (NDP).  This enables
          stateless address autoconfiguration in IPv6 clients on the
          network.
        '';
    };

    services.radvd.config = mkOption {
      example =
        ''
          interface eth0 {
            AdvSendAdvert on;
            prefix 2001:db8:1234:5678::/64 { };
          };
        '';
      description =
        ''
          The contents of the radvd configuration file.
        '';
    };

  };


  ###### implementation

  config = mkIf cfg.enable {

    users.extraUsers.radvd =
      { uid = config.ids.uids.radvd;
        description = "Router Advertisement Daemon User";
      };

    systemd.services.radvd =
      { description = "IPv6 Router Advertisement Daemon";
        wantedBy = [ "multi-user.target" ];
        after = [ "network.target" ];
        serviceConfig =
          { ExecStart = "@${pkgs.radvd}/bin/radvd radvd -n -u radvd -C ${confFile}";
            Restart = "always";
          };
      };

  };

}
{'fraction_non_alphanumeric': 0.11324093187850191, 'fraction_numerical': 0.012680625184311413, 'mean_word_length': 1.7986798679867986, 'pattern_counts': {'":': 0, '<': 7, '<?xml version=': 0, '>': 7, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1073792', 'n_tokens_mistral': 1066, 'n_tokens_neox': 984, 'n_words': 250}